

Using ODS Document with SAS/Graph to Remove Unwanted PDF Bookmarks

John Reilly, DataCeutics, Inc., Pottstown, PA

ABSTRACT

Using description=' ' with SAS Version 9.1 does not remove null bookmarks when using ODS PDF along with SAS/Graph as it does in SAS Version 8.2. See SN-011888. Using a null graph description does not remove the Table of Contents (TOC) entry/bookmarks. This paper describes a method to remove TOC bookmarks by using ODS Document with SAS/Graph in conjunction with ODS Output.

INTRODUCTION

A PDF document with multiple graphs in it was created using SAS 8.2 and I wanted to try it using SAS 9.1. I noticed right away that the output had a hierarchal display of bookmarks that it didn't have before. I was very confused as to why this would happen across versions of SAS, but a quick search turned up issue SN-011888 stated above. So I began my search for a work around and stumbled onto some information on ODS Document.

ODS Document is a very powerful tool that allows the user to create a file of output objects. These objects can be replayed in any of the ODS output formats without rerunning the program. ODS Document output is stored in a file called an Item Store. The item store file format permits client applications to define a "hierarchical file system within a file". It supports arbitrarily nested subdirectories, as well as arbitrarily sized binary files called *items*. Items are stored in compressed form, and are automatically uncompressed when read. The objects are given a sequence number as they are added to the Item Store, therefore SAS can identify output objects uniquely. ODS Document is not supported for PROC REPORT and it has limited support for PROC PRINT and PROC FREQ. All other Base SAS procedures are supported. ODS Output is not portable across operating environments.


I thought we could store the output into ODS Document and then rearrange it to get rid of some of the extra placeholders. The following example shows how easy it is to use ODS Document and the expected behavior when used with the DataCeutics CR Toolkit.

ODS DOCUMENT STATEMENT

First, a subset of the data was created named stat and then it was passed to the DataCeutics CR Toolkit Graph4 Macro. This macro calls the SAS/Graph Proc Gplot module. The DataCeutics CR Toolkit titles driver file controls where the output is placed, how it is named and what type is created. The unbolded portion of the below code generates the output in **Figure 1**. When wrapped with the ODS Document code, it also creates the Item Store called fpk1. Notice all of the bookmarks in the pane to the left; there are three bookmarks for each graph.

```
*****
* Create report using a CR Toolkit graph4 call.
*****;
ods document name=fpk1(write);
%graph4(in=stat,
        pageby=usubjid,
        id_var=trtnum,
        gtype=mean,
        v_var=mean,
        h_var=visit,
        offset=2,
        interpol=j);
ods document close;
```

Figure 1



ODS OUTPUT AND PROC DOCUMENT LIST STATEMENTS

The following code uses ODS Output to create a dataset with the structural information of the Item Store. This dataset, called `properties`, is renamed to `fpk1ds`. The Document Procedure also generates a contents file from the `fpk1` Item Store and prints them to an output file. **Figure 2** shows the dataset and each observation in it. Once the structure of the Item Store is in the dataset, it can be used to select only the items that are not of type "Dir". This leaves the "Graph" types in the dataset. By looking at **Figure 2** you can see that the path of the first graph is "\Gplot#1\ByGroup1#1\Gplot#1". A data Null step is used to turn this path into a macro variable which is later used in a do loop. A counting variable "zz" is also created. This is used to number the graphs by page.

```

/** Get the document information for bookmark removal */
ods output properties=fpk1ds;
  proc document name=fpk1;
    list / levels=all;
  run;
  quit;
ods output close;

data _null_; set fpk1ds(where=(type ne "Dir"));
  Call symput(compress("path"||_n_),trim(left(path)));
  Call symput("zz",trim(_n_));
run;

```

Figure2

Listing of: \Work.Fpk1\
Order by: Insertion
Number of levels: All

Obs	Path	Type
1	\Gplot#1	Dir
2	\Gplot#1\ByGroup1#1	Dir
3	\Gplot#1\ByGroup1#1\Gplot#1	Graph
4	\Gplot#2	Dir
5	\Gplot#2\ByGroup1#1	Dir
6	\Gplot#2\ByGroup1#1\Gplot1#1	Graph
7	\Gplot#3	Dir
8	\Gplot#3\ByGroup1#1	Dir
9	\Gplot#3\ByGroup1#1\Gplot2#1	Graph
10	\Gplot#4	Dir
11	\Gplot#4\ByGroup1#1	Dir
12	\Gplot#4\ByGroup1#1\Gplot3#1	Graph
13	\Gplot#5	Dir
14	\Gplot#5\ByGroup1#1	Dir
15	\Gplot#5\ByGroup1#1\Gplot4#1	Graph
16	\Gplot#6	Dir
17	\Gplot#6\ByGroup1#1	Dir
18	\Gplot#6\ByGroup1#1\Gplot5#1	Graph
19	\Gplot#7	Dir
20	\Gplot#7\ByGroup1#1	Dir
21	\Gplot#7\ByGroup1#1\Gplot6#1	Graph

GETTING THE TITLES

Instead of keeping the default “Gplot” bookmark labels, a macro variable was created from the title. The macro variable `fpk1` will be used to display the label for each bookmark with the page number concatenated onto it. For example the “Gplot” bookmark label for the first graph will be changed from “Gplot” to “Figure PK1 Individual plasma concentrations (unit) of << Analyte >> for Group A and Group B (<<Subject >>) Page 1” etc.


```
/**/ Get title2 from titles file ***/  
proc sql noprint;  
  select text into :fpk1 from dictionary.titles  
  where type='T' and number=2;  
quit;
```

PROC DOCUMENT STATEMENT

To get a new PDF document with the bookmarks removed, replay the individual graph output which is stored in the ODS Document `fpk1`, using PROC Document. First, PDF is turned on as the Destination and then PROC Document is used to replay the graphs in the Item Store to the PDF destination. **Doc** is the Item Store to be opened which is `fpk1`. Next, a do loop iterates through all of the macro variables. The graph is then **copied** to a temporary directory. **Setlabel** puts the value of the title and page into the bookmark label. Lastly, the graph is **replayed** from the temporary Directory into a PDF destination. The resulting PDF output is shown in **Figure 3**. There is now one bookmark for each graph.

```
/**/ Replay the document into pdf with bookmarks ***/  
/**/ removed and titles as bookmark text ***/  
ods pdf file="fpk1_all.pdf";  
  %macro doc;  
  proc document;  
    doc name=fpk1;  
    %do zx=1 %to &zz;  
      copy &&path&zx to \G1;  
      setlabel \G1 "%trim(&fpk1) Page &zx";  
      replay \G1;  
    %end;  
  quit;  
  %mend doc;  
  %doc;  
ods pdf close;
```

Figure 3



CONCLUSION

I wanted to learn more about the ODS Document and I wanted to develop a work around for a problem. I determined that the ODS Document is a very powerful way to perform post processing on output to produce a slightly different look without the need for a tool external to SAS. The output destination can be changed to any other destination to generate different types of output. For example, to generate RTF output, just substitute RTF for PDF in the above example see below. This will work a lot better when PROC Report is supported and reports can be generated to different destinations. I look forward to the release of SAS 9.2 when all Base SAS procedures should be supported. This will allow for the ODS Document to be used to create one file to multiple destinations using one master program no matter what procedure is used.

```

/** Replay the document into rtf */
ods rtf file="fpk1_all.rtf";
  %macro doc;
    proc document;
      doc name=fpk1;
      %do zx=1 %to &zz;
        replay &&path&zx;
      %end;
    run;
  quit;
  %mend doc;
%doc;
ods rtf close;

```

REFERENCES

SAS Institute Web Site: <http://www.sas.com/>

ODS Document Reference: <http://support.sas.com/rnd/base/topics/odsdocument/ref.html>

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

John Reilly
DataCeutics, Inc.
1610 Medical Drive
Pottstown, Pa 19464
(610) 970-2333
reillyj@dataceutics.com

DataCeutics, Inc. (<http://www.dataceutics.com>) is a SAS Alliance Partner and a CDISC member.



CR Toolkit is a registered trademark of DataCeutics, Inc.

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.